CSCI44B Programming Perl II

Lecture 6

4/10/2003

6 CGI Programming with CGI.pm Continued

6.1 Using the Same Code to Generate and Process Forms

In previous lectures, we sent data to a CGI program by using a static HTML query page with form elements that let us enter data.  Since we now know how to generate HTML forms with CGI.pm, it would be easier to generate a query page with CGI code.  In fact, we could have one CGI program generate the HTML form and the page produced with the form data.   Here is a simple CGI program to demonstrate how this can be done.

#!/usr/bin/perl -w

#

use strict;

use CGI::Pretty;

my $cgi = new CGI;

print $cgi->header();

#

# See if a first and last name have been set in the CGI data.

# If they both have, send a welcome screen with their name.  If neither

# have, send a form so the user can enter their name.

#

if( $cgi->param('first') and $cgi->param('last')) {

    my $first = ucfirst( lc( $cgi->param('first') ) );

    my $last = ucfirst( lc( $cgi->param('last') ) );

    print $cgi->start_html("Welcome");

    print $cgi->h1("Hello, $first $last");

} else {

    print $cgi->start_html(-title => "Enter your name");

    #

    # If the user only puts a first or last name, tell

    # them to enter both.

    #

    if( $cgi->param('first') or $cgi->param('last')) {

        print $cgi->center( font({-color=>'red'},

                              "You must enter a",

                              ($cgi->param('last') ? "first":"last"),

                              "name"

                               )

                            );

    }

    print &generate_form();

}

print $cgi->end_html();

exit;

# =====================================================================

# Generate a the form HTML asking for first and last name.

#

sub generate_form {

    my $form = $cgi->start_form({-method=>'GET'});

    $form .= $cgi->h1("Please enter your name:");

    $form .= $cgi->p("First name",

                     $cgi->textfield('first')

                     );

    $form .= $cgi->p("Last name",

                     $cgi->textfield('last')

                     );

    $form .= $cgi->p($cgi->submit());

    $form .= $cgi->end_form();

    return $form;

}
This program is called 6.1a.cgi.  You can access it at this URL:

http://linux.cschabot.org/~fellers/cgi-bin/lec6/6.1a.cgi

The first time you go here, the CGI program isn't receiving any data so it displays a query page.

If you look in the CGI code, you'll see an if-else block where the if statement is testing to see if the 'first' and 'last' CGI parameters have been set.  If they haven't, the code runs the else block.  In the else block, we call generate_form() that creates an HTML form to ask for the user's first and last name.

Type in 'John', and 'Smith' for the first and last name, respectively, then press the 'Submit' button and you'll see this.

Notice now that after you press the 'Submit' button, the same CGI program is being called but this time the query string, 'first=John&last=Smith', is appended to the URL.  Now the second time through the CGI program, the 'first' and 'last' parameters are set so we run the if in the if-else block is true.  Here we generate a page that welcomes the user that has entered their name in correctly.

6.2 Cookies

Cookies are a form of persistent data that a CGI program can store in a user's browser cache.  Cookies can be used for many reasons.  A common one is to store the identity of a user, so that a user who logs into a web site that requires registration doesn't have to re-login every time they visit the site.

Cookies are created roughly like this:

· A user visits the default web page of a site that requires a user to login to see the content.  This default web page is generated with a CGI program.

· The user enters in their username and password.  The data is sent to a CGI program that verifies the username and password.  If the login was successful, the CGI program then sends instructions to the user's browser to create a cookie in that user's browser cache.  These instructions also tell the browser what to name the cookie and what data to put into the cookie.  In this case, the CGI program tells the browser to put a subscriber identification code, unique to the user, in the cookie.

· As the user visits various pages in that web site, the browser sends the a copy of the cookie back to the server.  If a CGI program is creating the pages the user wants to see, the CGI program can check the subscriber identification code in the cookie.  If the code is ok, the user will see the requested page.  If not, the user can be redirected to a page that asks the user to subscribe.

· Furthermore, days or weeks later, when the user goes back to the default page of the same web site, the CGI program that runs the site will see that the user has a valid subscriber identification code and won't make to user login again.

Note that a cookie is stored on a specific user's computer.  If, in the above example, the user logged in, then walked to another computer and tried to access the site again from the second computer, he or she would have to login again because the second computer doesn't have the cookie with the identification code.

A cookie has 6 parameters: name, value, expires, domain, path, and secure.  For this lecture I'll only cover the first three.  Learning the other three are left as a reading assignment for those of you who want to become cookie experts.

· name: This is of the cookie.  You set the name and use it to retrieve the cookie. Every cookie must have a name.

· value: This is the value (data) the cookie contains.

· expires: This sets how long a browser should keep the cookie.  If this is not set, the default is to erase the cookie when the user exists the browser.  expires can be set as an interval (e.g. 5 days) or to a specific time and date.

Here is the minimal code to create a cookie called 'sampleCookie' with a value containing the string 'fubar':

my $cgi = new CGI;

my $cookie = $cgi->cookie(-name=>'sampleCookie',

                          -value=>'fubar');

print $cgi->header(-cookie=>$cookie);

The second line creates a cookie using the CGI.pm cookie() method.  The cookie isn't actually created in the user's browser until the third line, were we send it in the header of the HTML page being sent to the user's browser.

If the cookie already exists, we can retrieve the value by just naming the cookie in the cookie() method.

my $cookie_value = $cgi->cookie('sampleCookie');

If the cookie doesn't exists, $cookie_value would be undefined.  Cookie values can also take the form of arrays and hashes but for simplicity, we'll only address simple strings in this lecture.

In the following code, we take the code from §6.1 and modify it in two ways.  First, we change the query form so that is only asks for a username instead of a first and last name.  Second, after the user submits the username, the page returned by the CGI program also sets a cookie with this username as the value.  If we kill the browser, restart it, and go back to the CGI program, the CGI program retrieves the cookie's value and prints it in the browser's window.

#!/usr/bin/perl -w

#

# Cookie example

#
use strict;

use CGI;

my $cgi = new CGI;

#

# Try to read myCookie.  If undef is returned, cookie doesn't

# exist.

#
my $cookie_value = $cgi->cookie('myCookie');

#

# If there was a cookie or a username in the CGI parameters,

# make a HTML page welcoming the user.  Also tell the user

# if the username was found in a cookie or parameter.

# Both types of pages will also set a cookie or reset an

# existing cookie.

#

# If there is no cookie or parameter, present a login form.

#
if( $cookie_value || $cgi->param('userName') ) {

    my $cookie = $cgi->cookie(-name=>"myCookie",

                              -value=>$cgi->param('userName'),

                              -expires=>"+1d" # keep cookie 1 day
                             );

    print $cgi->header(-type=>"text/html", -cookie=>$cookie);

    print $cgi->start_html();

    if($cookie_value) {

        print $cgi->h1("Welcome $cookie_value!");

        print $cgi->h2("I recognize you from your cookie.");

    }else{

        print $cgi->h1("Welcome ",$cgi->param('userName'));

        print $cgi->h2("Thanks for logging in and setting ",

                       "your cookie.");

    }

    print $cgi->end_html();

}else{

    print $cgi->header();

    print $cgi->start_html();

    print generate_form();

    print $cgi->end_html();

}

sub generate_form {

    my $form = $cgi->start_form({-method=>'GET'});

    $form .= $cgi->h1("Please enter your user name:");

    $form .= $cgi->p("User Name:",

                     $cgi->textfield('userName')

                     );

    $form .= $cgi->p($cgi->submit({-value=>'ENTER'}));

    $form .= $cgi->end_form();

    return $form;

}

This CGI program is found here.

http://linux.cschabot.org/~fellers/cgi-bin/lec6/6.2a.cgi

When you first access this URL, there is no query string in the URL and there is no cookie in the browser's cache to identify yourself, so the CGI program servers a HTML form asking for your username.

In this example, I entered 'JohnSmith1029' as a username and pressed 'ENTER'.  The username is sent to the CGI program in the query string as shown below.  The CGI program receives this data and serves a web page acknowledging your username.  Additionally, the CGI program also sends a cookie in the header of this acknowledgement page.  This cookie is named 'myCookie' and contains the username, 'JohnSmith1029', as the value.

I then exit the browser and restart it again.  I then access the same CGI program as if I was starting over.  This time around, however, the CGI program finds 'myCookie', retrieves the value, and determines that I'm 'JohnSmith1029'.

If you want to repeat this example, you may have to delete an already existing cookie.  To do this in IE, go to Tools|Internet Options.  Click on the General tab and look for the Delete Cookies button and click it.  If you instead want to verify that the cookie is there, click on the Settings button instead.  In the Settings dialogue box click on View Files to see the files in the browser's cache.

Here's what my cache looks like.

There's one cookie, the one I created.  You can drag it to the desktop and open it with a text editor.  There you'll see the cookie name, value, the URL of the CGI program that created the cookie, and other encoded information.

